1 MRLocus statistical model

MRLocus proceeds in two separate hierarchical models, which are encoded in the Stan programming language and with posterior inference performed using the Stan and RStan software packages [Carpenter et al., 2017, Stan Development Team, 2020]. In section 1.1 we define the model for the colocalization step, and in section 1.2 we define the model for the slope fitting step.

1.1 Colocalization step

1.1.1 Input data

The first step performs colocalization of eQTL (A) and GWAS (B) signals across a number of nearly-LD-independent signal clusters \( j \in 1, \ldots, J \), using summary statistics from both studies: \( \hat{\beta}^A_{i,j} \) and \( \text{se}(\hat{\beta}^A_{i,j}) \), \( \hat{\beta}^B_{i,j} \) and \( \text{se}(\hat{\beta}^B_{i,j}) \) for SNP \( i \in 1, \ldots, n_j \) in cluster \( j \) for study A and B, and the respective LD matrices for each cluster \( j \): \( \Sigma^A_j \) and \( \Sigma^B_j \). Near LD independence is obtained by clumping with PLINK [Purcell et al., 2007] on the putative mediator A with \( r^2 \) threshold of 0.1 and then trimming “signal clusters” (the term we use for PLINK clumps after they are created), in which the pairwise \( r^2 \) of the index eSNP is greater than 0.05. Trimming is performed with the `trimClusters` function in MRLocus, which removes signal clusters with \( r^2 > 0.05 \) to the first cluster, then the second cluster, and so on until no pairs remain with \( r^2 > 0.05 \). The clusters and the individual SNPs per cluster must be matched across study. The estimated coefficients \( \hat{\beta}^X_{i,j} \) for \( X \in \{A, B\} \) refer to either the estimated coefficients from a linear model of a continuous trait \( y \) on genotype dosages \( \{0, 1, 2\} \), or the estimated log odds from a logistic regression of a binary trait \( y \) modeled on genotype dosages.

While it would be preferable to use allelic fold change (aFC) [Mohammadi et al., 2017] or ACME effect sizes [Palowitch et al., 2018] for the eQTL (A) study in MRLocus modeling, in practice we typically are provided with publicly available estimated coefficients representing inverse normal transformed (INT) or \( \log_2 \) transformed expression values regressed on genotype dosages. For eQTL coefficients derived from INT expression data, the mediation effect estimated by MRLocus represents the effect on the trait from modifying gene expression by 1 SD, while for eQTL coefficients derived from \( \log_2 \) transformed expression data, the mediation effect represents the effect on the trait from doubling gene expression.

The eQTL and GWAS studies are referred to as “A” and “B” in the formula and code below for generalization, for example, the eQTL study could be replaced with a pQTL (protein quantitative trait loci) study. “A” therefore refers to a study of a trait (or “exposure”) that is believed to be causally upstream of the trait examined in study “B” (or “outcome”).
1.1.2 Collapsing and allele flipping

MRLocus contains two convenience functions, `collapseHighCorSNPs` and `flipAllelesAndGather`, which are described briefly. The first function uses hierarchical clustering based on the LD matrix (the user must pick which to use if two are available), in order to collapse SNPs into groups using complete linkage, and thresholding the resulting dendrogram at 0.95 correlation. The SNP with the highest absolute z-score within a collapsed set is chosen as the representative SNP.

`flipAllelesAndGather` performs a number of allele flipping steps for assisting statistical modeling and visualization. The alleles are flipped such that the index SNP (as defined by its absolute value of z-score) for study A has a positive estimated coefficient. This is to simplify the interpretations of the plots – such that we are always describing the effects on downstream traits for expression increasing alleles. Additionally, we flip alleles such that SNPs with positive correlation of genotypes in either \( \Sigma^A_j \) or \( \Sigma^B_j \) (the user must pick which to use) are kept the same, while SNPs with negative correlation of genotypes have their alleles flipped. Allele flipping involves both keeping track of the reference and effect allele, as well as changing the sign of the estimated coefficient. This function also performs checks such that the A and B study agree in terms of the effect and reference allele. If two LD matrices are provided, one is prioritized for generating positive correlations of genotypes, while the other has its alleles flipped for consistency.

1.1.3 Scaling

In practice, before supplying \( \hat{\beta}^A_{i,j} \) and \( \hat{\beta}^B_{i,j} \) and the associated standard errors to the colocalization hierarchical model, the values are scaled such that the index SNP (as defined by its absolute value of z-score) for study A has estimated coefficient of \( \pm 1 \) for both study A and B. If two or more SNPs have the same z-score, the first is chosen. This simplifies the Stan code and improves model fit, as the two studies are then at comparable scale. The scaling is reversed after the Stan model is fit. For the user-input estimated coefficients and standard errors for study \( X \in \{A, B\} \) and cluster \( j \), \( \hat{\beta}^{X,\text{input}}_{i,j} \) and \( \text{se}(\hat{\beta}^{X,\text{input}}_{i,j}) \), in the first step we create scaled estimated coefficients:

\[
\begin{align*}
    z_j^* &= \max_i \left( |\hat{\beta}^{A,\text{input}}_{i,j}| / \text{se}(\hat{\beta}^{A,\text{input}}_{i,j}) \right) \\
    i_j^* &= \min \{i \in 1, \ldots, n_j\} \text{ s.t. } |\hat{\beta}^{A,\text{input}}_{i_j^*,j}| / \text{se}(\hat{\beta}^{A,\text{input}}_{i_j^*,j}) = z_j^* \\
    s_j^X &= 1/|\hat{\beta}^{X,\text{input}}_{i_j^*,j}| \\
    \hat{\beta}^{X}_{i,j} &= s_j^X \hat{\beta}^{X,\text{input}}_{i,j}, \ i \in 1, \ldots, n_j \\
    \text{se}(\hat{\beta}^{X}_{i,j}) &= s_j^X \text{se}(\hat{\beta}^{X,\text{input}}_{i,j}), \ i \in 1, \ldots, n_j
\end{align*}
\]

Note that equations (1-2) refer specifically to study A, while equations (3-5) refer to steps that are repeated for \( X = A \) and \( X = B \). Again, in words, \( i_j^* \) is the index of the first occurrence of the maximal value of absolute value of z-score in study A and cluster \( j \).

1.1.4 Colocalization

Colocalization refers to the task of determining if the same signal in eQTL and GWAS summary statistics arise from the same causal variant(s), considering the correlation of genotypes in a locus (the LD matrix). Here we perform colocalization using a generative model for the estimated coefficients, where the true coefficients will be modeled and their posterior distribution used for inference. In the following equations, \( \hat{\beta}^X_{i,j} \) and \( \text{se}(\hat{\beta}^X_{i,j}) \) refer to these scaled estimates as described
in the previous section, but in the following section 1.2 on slope fitting, the values are transformed back to the original scale by multiplying by $1/s_j^X$ for $X \in \{A, B\}$ respectively.

We use a statistical model for the summary statistics motivated by the eCAVIAR model [Hormozdiari et al., 2016]. In eCAVIAR, the summary statistic $z$-scores in a vector $S$ are modeled as a multivariate normal distribution with mean vector $\Sigma \Lambda$ and covariance matrix $\Sigma$ (the LD matrix), where $\Lambda$ is a vector giving the true standardized effect sizes. In a locus with a single causal SNP producing the observed enrichment of signal, and if the causal SNP is in the set modeled by eCAVIAR, $\Lambda$ would consist of a vector with all 0 values except for one SNP with non-zero value. eCAVIAR then models $\Lambda$ with a multivariate normal prior distribution centered on 0 with covariance matrix based on a vector of 0’s and 1’s giving the true causal status of the SNPs in the locus and a preset scale parameter determined from previous studies.

Here MRLocus diverges from eCAVIAR in two ways. First, we will use a generative model for the elements in the vector $\hat{\beta}_j^X$ conditional on the true effect sizes $\beta_{i,j}^X$ within nearly-LD-independent signal cluster $j$ (that is, MRLocus models effect sizes rather than $z$-scores). Second, we will use a univariate distribution for each $\hat{\beta}_{i,j}^X$ instead of a multivariate distribution for the entire vector. This second difference is motivated by practical concerns of model fitting and specification; the univariate modeling provided more efficient model fitting with Stan (higher effective sample sizes and R-hat values near 1), and allowed for more flexible choice of priors, as described below.

Instead of a multivariate normal prior on the estimated coefficients, MRLocus uses a horseshoe prior [Carvalho et al., 2009, 2010], a type of hierarchical shrinkage prior that provides sparsity in the posterior estimates [Piironen and Vehtari, 2017]. In the following, we use the notation of Carvalho et al. [2009], where $\lambda_i$ provides the local shrinkage parameters and $\tau$ provides the global shrinkage parameter.

The following hierarchical model is fit separately across nearly-LD-independent signal clusters $j \in 1, \ldots, J$, and so in the following equations, the subscript for $j$ is omitted for clarity. In all equations below but the last, $i \in 1, \ldots, n_j$. Here, for consistency with Stan code, the normal distribution is written as $N(\mu, \sigma)$ where the second element $\sigma$ provides the standard deviation instead of the variance.

\[
\hat{\beta}_i^A \sim \mathcal{N}(\Sigma^A \beta_i^A, se(\hat{\beta}_i^A)) \quad (6)
\]

\[
\hat{\beta}_i^B \sim \mathcal{N}(\Sigma^B \beta_i^B, se(\hat{\beta}_i^B)) \quad (7)
\]

\[
\beta_i^A \sim \mathcal{N}(0, \lambda_i \tau) \quad (8)
\]

\[
\beta_i^B \sim \mathcal{N}(0, \lambda_i \tau) \quad (9)
\]

\[
\lambda_i \sim \text{Cauchy}(0, 1) \quad (10)
\]

\[
\tau \sim \text{Cauchy}(0, 1) \quad (11)
\]

Of note, $\beta_i^A$ and $\beta_i^B$ share a prior involving $\lambda_i$, such that evidence from study A and B supporting a SNP $i^\dagger$ that is causal both for eQTL and GWAS signal ($\beta_i^{A^\dagger} \neq 0, \beta_i^{B^\dagger} \neq 0$), will be reflected in larger posterior draws for $\lambda_i$ compared to other $\lambda_{i'}$ for $i' \neq i^\dagger$.

The posterior mean for $\beta_{i,j}^A$ and $\beta_{i,j}^B$ are the parameters of interest from this step of model fitting, and passed along to the next step after scaling back to the original scale by $1/s_j^A$ and $1/s_j^B$, respectively, as described earlier. We also considered making use of the posterior standard deviation for these two parameters, but found MRLocus gave better performance in terms of accuracy and stability in the Stan fitting procedure if only the posterior mean was kept from the colocalization step. The use of the horseshoe prior in this colocalization step is distinct from other uses of the
horseshoe prior in Bayesian Mendelian Randomization methods, Berzuini et al. [2018] and Uche-Ikonne et al. [2019], where it is used as a prior for the pleiotropic effects (effects not mediated by the exposure) or on the mediation slope, respectively.

1.1.5 Colocalization with eCAVIAR (optional alternative)

Alternatively, MRLocus can accept colocalization results from eCAVIAR as input to the slope fitting step, as explored in the main text. eCAVIAR is run with default options on each nearly-LD-independent signal cluster separately, supplying the LD matrix, and z-scores for study A and B, and $-c 1$, i.e. setting the maximum number of causal SNPs to 1. The following section describes how the output from eCAVIAR colocalization can be used alternatively to the MRLocus colocalization output.

1.2 Slope fitting step

The second step of MRLocus is to estimate the gene-to-trait effect $\alpha$ (the slope in a Mendelian Randomization analysis) using the posterior mean values from the colocalization step. For each nearly-LD-independent signal cluster $j$, MRLocus extracts one SNP, based on the largest posterior mean value for $\beta^{A}_{i,j}$ (prioritizing the putative mediator A for SNP selection). MRLocus also has the non-default option to perform model-based clustering using an EM algorithm to determine how many SNPs per nearly-LD-independent signal cluster to pass to the slope fitting step [Scrucca et al., 2016], but this option was not extensively evaluated here. For eCAVIAR colocalization, the SNP with the largest colocalization posterior probability (CLPP) is selected from each signal cluster, among those SNPs which are valid instruments, i.e. with an absolute value of z-score as large as the value used for PLINK clumping [Purcell et al., 2007]. eCAVIAR alleles are flipped such that the alternate allele is the one corresponding to an increase in the measured phenotype in A (e.g. gene expression). Another round of $r^2$ trimming is performed, as in section 1.1.1, so that all chosen SNPs for slope fitting have pairwise $r^2 < 0.05$.

By default, $J$ SNPs are passed from the colocalization step to the slope fitting step, which should represent the candidate causal SNPs from each nearly-LD-independent signal cluster. The posterior mean for $\beta^{A}_{i,j}$ and $\beta^{B}_{i,j}$ for these SNPs is then provided to the slope fitting step as variables $\hat{\beta}^{A}_{j}$ and $\hat{\beta}^{B}_{j}$. These two variables are modeled as normally distributed variables centered on true values $\beta^{A}_{j}$ and $\beta^{B}_{j}$ with standard deviation according to the original standard errors $se(\hat{\beta}^{A}_{i,j})$ and $se(\hat{\beta}^{B}_{i,j})$. We found this incorporation of the original standard errors into the slope fitting procedure helped to accurately estimate the uncertainty on the slope (the gene-to-trait effect).

The primary two parameters of interest are the slope $\alpha$, i.e. the predominant gene-to-trait effect demonstrated by the nearly-LD-independent signal clusters in the locus, and $\sigma$, the dispersion of gene-to-trait effects from individual signal clusters around the predominant slope. The hierarchical model for the slope fitting step is given by:
\[ \hat{\beta}_j^A \sim N(\hat{\beta}_j^A, se_j^A) \] (12)
\[ \hat{\beta}_j^B \sim N(\hat{\beta}_j^B, se_j^B) \] (13)
\[ \beta_j^A \sim N(0, SD_\beta) \] (14)
\[ \beta_j^B \sim N(\alpha\beta_j^A, \sigma) \] (15)
\[ \alpha \sim N(\mu_\alpha, SD_\alpha) \] (16)
\[ \sigma \sim HN(0, SD_\sigma) \] (17)

where the first four equations are defined for \( j \in 1, \ldots, J \), and where \( HN \) specifies the Half-Normal distribution. In words, \( \beta_j^B \) is assumed to follow a normal distribution centered on the predicted value from a line with slope \( \alpha \) and no intercept, and with dispersion \( \sigma \) around the fitted line. We are both interested in the posterior mean of \( \alpha \) and \( \sigma \) as well as our uncertainty regarding the point estimates. In particular we focus on a quantile-based credible interval for \( \alpha \), which is used for determining our confidence in a gene being a causal mediator for a trait. Large values of \( \sigma \) relative to \( \alpha \) times typical mediator perturbation sizes (\( \beta_j^A \)) reflect significant heterogeneity of the gene-to-trait effect.

### 1.2.1 Choice of hyperparameters

The hyperparameter for the prior for \( \beta_j^A \) is \( SD_\beta \), which is set to 2 times the largest absolute value of \( \hat{\beta}_j^A \). The hyperparameters for the prior for \( \alpha \) are \( \mu_\alpha \) and \( SD_\alpha \), and are set based on a simple un-weighted linear model of \( \hat{\beta}_j^B \) on \( \hat{\beta}_j^A \) without an intercept. \( \mu_\alpha \) is set to the estimated slope coefficient, and \( SD_\alpha \) is set to 2 times the absolute value of this estimated slope coefficient.

The hyperparameter for the prior for \( \sigma \) is \( SD_\sigma \), which is set to \( \max(2 \text{sd}(\hat{\beta}_j^B), \max_j(|\hat{\beta}_j^B|)) \), where \( \text{sd}(\cdot) \) denotes the sample standard deviation. That is, the larger value between (i) 2 times the standard deviation of the coefficients for \( B \) and (ii) the largest absolute value of coefficient for \( B \). When the coefficients for \( B \) are widely dispersed, then taking two times the SD results in a wide prior for \( \sigma \), and when the coefficients are not widely dispersed, then taking the absolute value of the largest coefficient ensures the prior width for \( \sigma \) does not become too small.

Prior predictive checks [Gabry et al., 2019], in which data is generated from the estimated prior using the data generating mechanism, are provided in the MRLocus package and demonstrated in the vignette.

### 1.2.2 Loci with only one signal cluster (\( J = 1 \))

It is not recommended to use MRLocus with only one signal cluster.

### 1.2.3 Variation in \( \alpha \) in simulations

We note that the value for true slope parameter \( \alpha \) (x-axis in the simulation assessment plots) varies across iterations in the simulation for fixed h^2med and h^2g. The reason for this variation is that the \( \alpha \) provided by the twas_sim simulation framework corresponds to the gene-to-trait mediated effect, when eQTL and GWAS effect sizes are standardized using the SD of gene expression and trait in the eQTL and GWAS samples (as would occur in real data analysis). The h^2g and h^2med in the twas_sim simulator describe the population-level expression heritability and mediated trait
heritability. As the eQTL and GWAS sample sizes grow (e.g. already seen in the eQTL $N = 1,000$ simulation), the variation is reduced and we have $\alpha$ values converging to $\sqrt{\frac{h^{2\text{med}}}{h^2_g}}$. 
2 MRLocus Stan code

2.1 Colocalization step

```
data {
  int n;
  vector[n] beta_hat_a;
  vector[n] beta_hat_b;
  vector[n] se_a;
  vector[n] se_b;
  matrix[n,n] Sigma_a;
  matrix[n,n] Sigma_b;
}
parameters {
  vector[n] beta_a;
  vector[n] beta_b;
  vector<lower=0>[n] lambda;
  real<lower=0> tau;
}
model {
  tau ~ cauchy(0, 1);
  lambda ~ cauchy(0, 1);
  beta_hat_a ~ normal(Sigma_a * beta_a, se_a);
  beta_hat_b ~ normal(Sigma_b * beta_b, se_b);
  for (i in 1:n) {
    beta_a[i] ~ normal(0, lambda[i] * tau);
    beta_b[i] ~ normal(0, lambda[i] * tau);
  }
}
2.2 Slope fitting step

inst/stan/slope.stan

data {
  int n;
  vector[n] beta_hat_a;
  vector[n] beta_hat_b;
  vector[n] sd_a;
  vector[n] sd_b;
  real sd_beta;
  real mu_alpha;
  real sd_alpha;
  real sd_sigma;
}

parameters {
  real alpha;
  real<lower=0> sigma;
  vector[n] beta_a;
  vector[n] beta_b;
}

model {
  beta_hat_a ~ normal(beta_a, sd_a);
  beta_hat_b ~ normal(beta_b, sd_b);
  beta_a ~ normal(0, sd_beta);
  beta_b ~ normal(alpha * beta_a, sigma);
  alpha ~ normal(mu_alpha, sd_alpha);
  sigma ~ normal(0, sd_sigma);
}
References

Bob Carpenter, Andrew Gelman, Matthew Hoffman, Daniel Lee, Ben Goodrich, Michael Betancourt, Marcus Brubaker, Jiqiang Guo, Peter Li, and Allen Riddell. Stan: A probabilistic programming language. *Journal of Statistical Software, Articles*, 76(1):1–32, 2017. doi:10.18637/jss.v076.i01.

Stan Development Team. RStan: the R interface to Stan, 2020. URL http://mc-stan.org/. R package version 2.21.1.

Shaun Purcell, Benjamin Neale, Kathe Todd-Brown, Lori Thomas, Manuel A.R. Ferreira, David Bender, Julian Maller, Pamela Sklar, Paul I.W. de Bakker, Mark J. Daly, and Pak C. Sham. PLINK: a toolset for whole-genome association and population-based linkage analysis. *American Journal of Human Genetics*, 81(3), 2007. doi:10.1086/519795.

Pejman Mohammadi, Stephane E. Castel, Andrew A. Brown, and Tuuli Lappalainen. Quantifying the regulatory effect size of cis-acting genetic variation using allelic fold change. *Genome Research*, 27(11):1872–1884, 2017. doi:10.1101/gr.216747.116.

John Palowitch, Andrey Shabalim, Yi-Hui Zhou, Andrew B. Nobel, and Fred A. Wright. Estimation of cis-eqtl effect sizes using a log of linear model. *Biometrics*, 74(2):616–625, 2018. doi:10.1111/biom.12810.

Farhad Hormozdiari, Martijn van de Bunt, Ayellet V. Segrè, Xiao Li, Jong Wha J. Joo, Michael Bilow, Jae Hoon Sul, Sriram Sankararaman, Bogdan Pasaniuc, and Eleazar Eskin. Colocalization of gwas and eqtl signals detects target genes. *The American Journal of Human Genetics*, 99(6):1245–1260, 2016. doi:10.1016/j.ajhg.2016.10.003.

Carlos M. Carvalho, Nicholas G. Polson, and James G. Scott. Handling sparsity via the horseshoe. *Proceedings of Machine Learning Research*, 5:73–80, 2009. URL http://proceedings.mlr.press/v5/carvalho09a.html.

Carlos M. Carvalho, Nicholas G. Polson, and James G. Scott. The horseshoe estimator for sparse signals. *Biometrika*, 97(2):465–480, 2010. doi:10.1093/biomet/asq017.

Juho Piironen and Aki Vehtari. On the Hyperprior Choice for the Global Shrinkage Parameter in the Horseshoe Prior. *Proceedings of Machine Learning Research*, 54:905–913, 2017. URL http://proceedings.mlr.press/v54/piironen17a.html.

Carlo Berzuini, Hui Guo, Stephen Burgess, and Luisa Bernardinelli. A Bayesian approach to Mendelian randomization with multiple pleiotropic variants. *Biostatistics*, 21(1):86–101, 2018. doi:10.1093/biostatistics/kxy027.

Okezie O Uche-Ikonne, Frank Dondelinger, and Tom Palmer. Bayesian estimation of IVW and MR-Egger models for two-sample Mendelian randomization studies. *medRxiv*, 2019. doi:10.1101/19005868.

Luca Scrucca, Michael Fop, T. Brendan Murphy, and Adrian E. Raftery. mclust 5: clustering, classification and density estimation using Gaussian finite mixture models. *The R Journal*, 8(1):289–317, 2016. doi:10.32614/RJ-2016-021.
Jonah Gabry, Daniel Simpson, Aki Vehtari, Michael Betancourt, and Andrew Gelman. Visualization in Bayesian workflow. *Journal of the Royal Statistical Society: Series A (Statistics in Society)*, 182(2):389–402, 2019. doi:10.1111/rss.a.12378.